




Many of the slides in this lecture are either from or adapted from slides provided by the authors of the textbook "Computer Systems: A Programmer's Perspective," 2nd Edition and are provided from the website of Carnegie-Mellon University, course 15-213, taught by Randy Bryant and David O'Hallaron in Fall 2010. These slides are indicated "Supplied by CMU" in the notes section of the slides.








Procedure Control Flow


- Use stack to support procedure call and return
- Procedure call: call sub
 - push return address on stack
 - jump to sub
- Return address:
 - address of the next instruction after call
 - example from disassembly


```
804854e: e8 3d 06 00 00 call 8048b90 <sub>
8048553: 50 pushl %eax
```


- return address = 0x8048553
- Procedure return: ret
 - pop address from stack
 - jump to address


CS33 Intro to Computer Systems XII-5














The *leave* instruction causes the contents of ebp to be copied into esp, thereby removing everything from the stack that had been pushed into the frame. It then pops the current stack top (the old ebp) into the ebp register. The effect of *leave* is thus to return to the caller's stack frame.

There is an *enter* instruction that has the same effect as that of the first three instructions of subr combined (it has an operand that indicates how much space for local variables to allocate). However, it's not used by gcc, apparently because it's slower than doing it as shown in the slide.

Register-Saving Conventions

- When procedure yoo calls who:
 - yoo is the caller
 - who is the callee
- · Can registers be used for temporary storage?

```
yoo:

movl $33, %edx
call who
addl %edx, %eax

ret
```

```
who:

movl 8(%ebp), %edx
addl $32, %edx

ret
```

- contents of register %edx overwritten by who
- $\boldsymbol{-}$ this could be trouble: something should be done!

» need some coordination

CS33 Intro to Computer Systems


XII-12

Register-Saving Conventions

- When procedure yoo calls who:
 - yoo is the caller
 - who is the callee
- Can registers be used for temporary storage?
- Conventions
 - "caller save"
 - » caller saves temporary values on stack before the call
 - » restores them after call
 - "callee save"
 - » callee saves temporary values on stack before using
 - » restores them before returning

CS33 Intro to Computer Systems

XII-13



```
Register-Saving Example
  yoo:
                               who:
                                  pushl %ebx
     movl $33, %edx
     pushl %edx
                                  . . .
                                  movl 4(%ebp), %ebx
     call who
     popl %edx
                                  addl %53, %ebx
     addl %edx, %eax
                                  movl 8(%ebp), %edx
                                  addl $32, %edx
     ret
                                  popl %ebx
                                  ret
CS33 Intro to Computer Systems
                            XII-15
```


```
Recursive Function
                                   pcount_r:
                                       pushl %ebp
                                       movl %esp, %ebp
/* Recursive popcount */
                                       pushl %ebx
int pcount_r(unsigned x) {
                                       subl $4, %esp
  if (x == 0)
                                       movl 8(%ebp), %ebx
    return 0;
                                       movl $0, %eax
  else return
                                       testl %ebx, %ebx
    (x \& 1) + pcount_r(x >> 1);
                                        je .L3
                                       movl %ebx, %eax
                                        shrl $1, %eax
                                       movl
                                             %eax, (%esp)


    Registers


                                       call pcount_r
                                       movl %ebx, %edx
    - %eax, %edx used without
     first saving
                                       andl $1, %edx
                                       leal (%edx,%eax), %eax

    - %ebx used, but saved at

                                    .L3:
     beginning & restored at end
                                        addl
                                              $4, %esp
                                       popl
                                             %ebx
                                       popl %ebp
                                        ret
 CS33 Intro to Computer Systems
                              XII-16
```







```
Recursive Call #4
/* Recursive popcount */
int pcount_r(unsigned x) {
                                                  %ebx, %edx
                                          movl
  if (x == 0)
                                          andl $1, %edx
    return 0;
                                          leal (%edx,%eax), %eax
  else return
    (x \& 1) + pcount_r(x >> 1);

    Assume


     - %eax holds value from recursive call
     - %ebx holds x
                                            %ebx x

    Actions

     - compute (x & 1) + computed value

    Effect

     - %eax set to function result
 CS33 Intro to Computer Systems
                                 XII-20
```





Observations About Recursion

- Handled without special consideration
 - stack frames mean that each function call has private storage
 - » saved registers & local variables
 - » saved return pointer
 - register-saving conventions prevent one function call from corrupting another's data
 - stack discipline follows call / return pattern
 - » if P calls Q, then Q returns before P
 - » last-in, first-out
- · Also works for mutual recursion
 - P calls Q; Q calls P

CS33 Intro to Computer Systems

XII-22





Why Bother with a Frame Pointer?

- It points to the beginning of the stack frame
 - making it easy for people to figure out where things are in the frame
 - but people don't execute the code ...
- The stack pointer always points somewhere within the stack frame
 - it moves about, but the compiler knows where it is pointing
 - » a local variable might be at 8(%rsp) for one instruction, but at 16(%rsp) for a subsequent one
 - » tough for people, but easy for the compiler
- · Thus the frame pointer is superfluous
 - it can be used as a general-purpose register

CS33 Intro to Computer Systems

XII-25

Copyright © 2015 Thomas W. Doeppner. All rights reserved.

Note that "frame pointer" is synonymous with "base pointer".

%rax	Return value	%r8	Argument #5
orax	Return value	210	Algument #3
%rbx	Callee saved	%r9	Argument #6
%rcx	Argument #4	%r10	Caller saved
%rdx	Argument #3	%r11	Caller Saved
%rsi	Argument #2	%r12	Callee saved
%rdi	Argument #1	%r13	Callee saved
%rsp	Stack pointer	%r14	Callee saved
%rbp	Callee saved	%r15	Callee saved

x86-64 Registers

- Arguments passed to functions via registers
 - if more than 6 integral parameters, then pass rest on stack
 - these registers can be used as caller-saved as well
- All references to stack frame via stack pointer
 - eliminates need to update %ebp/%rbp
- · Other registers
 - 6 callee-saved
 - 2 caller-saved
 - 1 return value (also usable as caller-saved)
 - 1 special (stack pointer)


CS33 Intro to Computer Systems

XII-27

Supplied by CMU.

Note that the *leave* instruction is no longer relevant, since %rbp does not contain the address of the stack frame.

Also note that the conventions shown in the slide are those adopted by gcc on Linux; they aren't necessarily used by other compilers or on other operating systems. Even gcc doesn't use these conventions if optimization is completely turned off (in which case arguments are passed on the stack, just as for IA32).



```
x86-64 Locals in the Red Zone
                                        swap a:
/* Swap, using local array */
                                          movq (%rdi), %rax
void swap a(long *xp, long *yp)
                                          movq %rax, -24(%rsp)
                                          movq (%rsi), %rax
    volatile long loc[2];
                                          movq %rax, -16(%rsp)
    loc[0] = *xp;
                                                -16(%rsp), %rax
                                          movq
    loc[1] = *yp;
                                          movq %rax, (%rdi)
     *xp = loc[1];
                                                -24(%rsp), %rax
                                          movq
     *yp = loc[0];
                                          movq %rax, (%rsi)
                                          ret

    Avoiding stack-pointer change

                                              rtn Addr
                                                           %rsp

    can hold all information within small

                                               unused
     window beyond stack pointer
                                              loc[1]
                                          -16
       » 128 bytes
                                              loc[0]
CS33 Intro to Computer Systems
                               XII-29
```

The *volatile* keyword tells the compiler that it may not perform optimizations on the associated variable such as storing it strictly in registers and not in memory. It's used primarily in cases where the variable might be modified via other routines that aren't apparent when the current code is being compiled. We'll see useful examples of its use later. Here it's used simply to ensure that *loc* is allocated on the stack, thus giving us a simple example of using local variables stored on the stack.

The issue here is whether a reference to memory beyond the current stack (as delineated by the stack pointer) is a legal reference. On IA32 it is not, but on x86-64 it is, as long at the reference is not more than 128 bytes beyond the end of the stack.

x86-64 NonLeaf without Stack Frame

```
· No values held while swap being
 /* Swap a[i] & a[i+1] */
                                        invoked
void swap ele(long a[], int i)
                                      · No callee-save registers needed
     swap(&a[i], &a[i+1]);

    rep instruction inserted as no-op

    based on recommendation from AMD

                                              » can't handle transfer of control to ret
  swap_ele:
                                          # Sign extend i
      movslq %esi,%rsi
      leaq
              8(%rdi,%rsi,8), %rax # &a[i+1]
      leaq
               (%rdi,%rsi,8), %rdi # &a[i] (1st arg)
              %rax, %rsi
                                          # (2<sup>nd</sup> arg)
      pvom
      call
              swap
      rep
                                          # No-op
      ret
CS33 Intro to Computer Systems
                                 XII-30
```

Supplied by CMU.

The *movslq* instruction copies a long into a quad, propagating the sign bit into the upper 32 bits of the quad word. For example, suppose %esi contains 0x08888888. After the execution of *movslq %esi*, %rsi, %rsi will contain 0x0000000088888888. But if %esi initially contains 0x88888888 (i.e., the sign bit is set), then after execution of the instruction, %rsi will contain oxffffffff888888888.

x86-64 Stack Frame Example

```
long sum = 0;
/* Swap a[i] & a[i+1] */
void swap_ele_su
   (long a[], int i)
{
     swap(&a[i], &a[i+1]);
     sum += (a[i]*a[i+1]);
}
```

- Keeps values of &a[i] and &a[i+1] in callee-save registers
 - rbx and rbp
- Must set up stack frame to save these registers
 - else clobbered in swap

CS33 Intro to Computer Systems


```
swap_ele_su:
           %rbx, -16(%rsp)
   movq
           %rbp, -8(%rsp)
   movq
   subq
           $16, %rsp
   movslq
           %esi,%rax
           8(%rdi,%rax,8), %rbx
   leaq
           (%rdi,%rax,8), %rbp
   leag
           %rbx, %rsi
   movq
           %rbp, %rdi
   movq
   call
           swap
           (%rbx), %rax
   movq
   imulq
           (%rbp), %rax
           %rax, sum(%rip)
   addq
   movq
           (%rsp), %rbx
           8(%rsp), %rbp
   movq
   addq
           $16, %rsp
   ret
```

XII-31

Supplied by CMU.

Note that sum is a global variable. While its exact location in memory is not known by the compiler, it will be stored in memory at some location just beyond the end of the executable code (which is known as "text"). Thus the compiler can refer to sum via the instruction pointer. The actual displacement, i.e., the distance from the current target of the instruction pointer and the location of sum, is not known to the compiler, but will be known to the linker, which will fill this displacement in when the program is linked. This will all be explained in detail in two or three weeks.

Understanding x86-64 Stack Frame swap_ele_su: movq %rbx, -16(%rsp) # Save %rbx %rbp, -8(%rsp) # Save %rbp \$16, %rsp # Allocate stack frame %osi %rax # Extend i into quad word movq subq movslq %esi,%rax leaq 8(%rdi,%rax,8), %rbx # &a[i+1] (callee save) leaq (%rdi,%rax,8), %rbp # &a[i] (callee save) movq %rbx, %rsi # 2nd argument movq %rbp, %rdi # 1st argument call swap movq (%rbx), %rax # Get a[i+1] imulq (%rbp), %rax # Multiply by a[i] addq %rax, sum(%rip) # Add to sum movq (%rsp), %rbx # Restore %rbx movq 8(%rsp), %rbp # Restore %rbp addq \$16, %rsp # Deallocate frame addq \$16, %rsp # Deallocate frame ret **CS33 Intro to Computer Systems** XII-32



Quiz 2

```
swap_ele_su:
         %rbx, -16(%rsp)
  movq
  movq
          %rbp, -8(%rsp)
          $16, %rsp
  subq
  movslq %esi,%rax
          8(%rdi,%rax,8), %rbx
  leaq
          (%rdi,%rax,8), %rbp
  leaq
  movq
          %rbx, %rsi
  movq
          %rbp, %rdi
  call
          swap
         (%rbx), %rax
  movq
  imulq
         (%rbp), %rax
  addq
          %rax, sum(%rip)
  movq
          (%rsp), %rbx
  movq
          8(%rsp), %rbp
          $16, %rsp
  addq
  ret
```

Since a 128-byte red zone is allowed, is it necessary to allocate the stack frame by subtracting 16 from %rsp?

- a) yes
- b) no

```
# Add to sum
# Restore %rbx
# Restore %rbp
# Deallocate frame
```

CS33 Intro to Computer Systems

XII-34

Copyright © 2015 Thomas W. Doeppner. All rights reserved.

Interesting Features of Stack Frame

- Allocate entire frame at once
 - all stack accesses can be relative to %rsp
 - do by decrementing stack pointer
 - can delay allocation, since safe to temporarily use red zone
- Simple deallocation
 - increment stack pointer
 - no base/frame pointer needed

CS33 Intro to Computer Systems

XII-35

x86-64 Procedure Summary


- Heavy use of registers
 - parameter passing
 - more temporaries since more registers
- Minimal use of stack
 - sometimes none
 - allocate/deallocate entire block
- Many tricky optimizations
 - what kind of stack frame to use
 - various allocation techniques

CS33 Intro to Computer Systems

XII-36

```
Tail Recursion
                                  int factorial(int x) {
int factorial(int x) {
  if (x == 1)
                                     return f2(x, 1);
    return x;
  else
                                   int f2(int a1, int a2) {
    return
       x*factorial(x-1);
                                     if (a1 == 1)
                                        return a2;
}
                                      else
                                        return
                                          f2(a1-1, a1*a2);
CS33 Intro to Computer Systems
                               XII-37
                                     Copyright © 2015 Thomas W. Doeppner. All rights reserved.
```

The slide shows two implementations of the factorial function. Both use recursion. In the version on the left, the result of each recursive call is used within the invocation that issued the call. In the second, the result of each recursive call is simply returned. This is known as *tail recursion*.



Here we look at the stack usage for the version without tail recursion. Note that we have as many stack frames as the value of the argument; the results of the calls are combined after the stack reaches its maximum size.

No Tail Recursion (2)

x: 6 return addr x: 5 return addr x: 4 return addr x: 3 return addr x: 2 return addr x: 1 return addr

ret: 720

ret: 120

ret: 24

ret: 6


ret: 2

ret: 1

CS33 Intro to Computer Systems

XII-39

Copyright © 2015 Thomas W. Doeppner. All rights reserved.



With tail recursion, since the result of the recursive call is not used by the issuing stack frame, it's possible to reuse the issuing stack frame to handle the recursive invocation. Thus rather than push a new stack frame on the stack, the current one is written over. Thus the entire sequence of recursive calls can be handled within a single stack frame.

```
Code: gcc -O1
     f2:
                        %esi, %eax
               movl
               cmpl
                        $1, %edi
                         .L5
               jе
                        $8, %rsp
               subq
               movl
                        %edi, %esi
               imull
                        %eax, %esi
               subl
                        $1, %edi
               call
                        f2
                                    # recursive call!
               addq
                         $8, %rsp
     .L5:
               rep
               ret
CS33 Intro to Computer Systems
                               XII-41
                                      Copyright © 2015 Thomas W. Doeppner. All rights reserved.
```

This is the result of compiling the tail-recursive version of factorial using gcc with the – O1 flag. This flags turns on a moderate level of code optimization, but not enough to cause the stack frame to be reused.

```
Code: gcc -O2
     f2:
                          $1, %edi
                cmpl
                movl
                           %esi, %eax
                           .L8
                iе
     .L12:
                imull
                           %edi, %eax
                subl
                          $1, %edi
                                                 loop!
                           $1, %edi
                cmpl
                jne
                           .L12
     .L8:
                rep
                ret
CS33 Intro to Computer Systems
                                 XII-42
                                         Copyright © 2015 Thomas W. Doeppner. All rights reserved.
```

Here we've compiled the program using the -O2 flag, which turns on additional optimization (at the cost of increased compile time), with the result that the recursive calls are optimized away — they are replaced with a loop.

Why not always compile with -O2? For "production code" that is fully debugged (assuming this is possible), this is a good idea. But this and other aggressive optimizations make it difficult to relate the runtime code with the source code. Thus, a runtime error might occur at some point in the program's execution, but it is impossible to determine exactly which line of the source code was in play when the error occurred.